```
#Chuong trinh: ham khong la
# Ham range, max_day, min_day
#---------------------------------
#Data segment
     .data
#Cac dinh nghia bien
int_arr:  .word     24,79,13,80,46
     .word     35,68,12,91,57
int_n:    .word     10
int_max:  .word     10
int_min:  .word     11
int_ran:  .word     13
#Cac cau nhac nhap du lieu
Xuat_max: .asciiz   "Gia tri lon nhat: "
Xuat_min: .asciiz   "Gia tri nho nhat: "
Xuat_ran: .asciiz   "Range = "
#---------------------------------
#Code segment
     .text
     .globl    main
#---------------------------------
#Chuong trinh chinh
#---------------------------------
main:
#Nhap (syscall)
#Xu ly
  # goi ham range
     la   $a0,int_arr
     lw   $a1,int_n
     jal  range
     sw   $v0,int_ran
#Xuat ket qua (syscall)
  # xuat max
     la   $a0,Xuat_max
     addi $v0,$zero,4
     syscall
     lw   $a0,int_max
     addi $v0,$zero,1
     syscall
  # xuong dong
     addi $a0,$zero,'\n'
     addi $v0,$zero,11
     syscall
  # xuat min
     la   $a0,Xuat_min
     addi $v0,$zero,4
     syscall
     lw   $a0,int_min
     addi $v0,$zero,1
     syscall
  # xuong dong
     addi $a0,$zero,'\n'
     addi $v0,$zero,11
     syscall
```

```
  # xuat range
     la   $a0,Xuat_ran
     addi $v0,$zero,4
     syscall
     lw   $a0,int_ran
     addi $v0,$zero,1
     syscall
#ket thuc chuong trinh (syscall)
Kthuc:    addiu     $v0,$zero,10
     syscall
#-----------------------------------
# Ham range:
# Input: $a0=addr(a[0]), $a1=so phan tu n
# Output: $v0=range
# Reserved: $ra
#-----------------------------------
range:
     subi $sp,$sp,4
     sw   $ra,0($sp)
#-----------------------------------
     jal  max_day
     sw   $v0,int_max
     jal  min_day
     sw   $v0,int_min
     lw   $t0,int_max
     sub  $v0,$t0,$v0
#-----------------------------------
     lw   $ra,0($sp)
     addi $sp,$sp,4
     jr   $ra
#-----------------------------------
# Ham max_day:
# Input: $a0=addr(a[0]), $a1=so phan tu n
# Output: $v0=phan tu lon nhat
# Reserved: $a0
#-----------------------------------
max_day:  sub  $sp,$sp,4
     sw   $a0,0($sp)
  # s0=a[0]/max, s1=a[i], s2=i(=1), s3=a[i]-max
     lw   $s0,0($a0)
  # for1 - init
     addi $s2,$zero,1
     addi $a0,$a0,4
  # cond1 i==n -> ket thuc
cond1:
     beq  $s2,$a1,endfor1
  # body1
    # if1 (a[i]>max) max=a[i]
     lw   $s1,0($a0)
     sub  $s3,$s1,$s0
     bltz $s3,endif1
    # then1: max=a[i]
     add  $s0,$s1,$zero
    # endif1
```

```
endif1:
  # loop1
     addi $s2,$s2,1
     addi $a0,$a0,4
     j    cond1
  # endf1
endfor1:
     add  $v0,$s0,$zero
     lw   $a0,0($sp)
     add  $sp,$sp,4
     jr   $ra
#-----------------------------------
# Ham min_day:
# Input: $a0=addr(a[0]), $a1=so phan tu n
# Output: $v0=phan tu nho nhat
# Reserved: $a0
#-----------------------------------
min_day:  sub  $sp,$sp,4
     sw   $a0,0($sp)
  # s0=a[0]/min, s1=a[i], s2=i(=1), s3=a[i]-min
     lw   $s0,0($a0)
  # for2 - init
     addi $s2,$zero,1
     addi $a0,$a0,4
  # cond2 i==n -> ket thuc
cond2:
     beq  $s2,$a1,endfor2
  # body2
    # if2 (a[i]<min) min=a[i]
     lw   $s1,0($a0)
     sub  $s3,$s1,$s0
     bgtz $s3,endif2
    # then2: max=a[i]
     add  $s0,$s1,$zero
    # endif2
endif2:
  # loop2
     addi $s2,$s2,1
     addi $a0,$a0,4
     j    cond2
  # endf2
endfor2:
     add  $v0,$s0,$zero
     lw   $a0,0($sp)
     add  $sp,$sp,4
     jr   $ra
#-----------------------------------
```