```
# Chuong trinh: C31B3 day so thuc
# ham nhap_pt, xuat_pt, max_day, min_day
#---------------------------------
# Data segment
     .data
# Cac dinh nghia bien
flo_arr:  .space    20        # day so thuc 5 phan tu
int_n:    .word     5         # so phan tu (spt)
flo_max:  .float    1.3
flo_min:  .float    1.1
# Cac cau nhac nhap du lieu
Nhap_day: .asciiz   "Nhap phan tu day:\n"
Nhap_pt1: .asciiz   "  a["
Nhap_pt2: .asciiz   "]: "
Xuat_day: .asciiz   "Day da nhap:\n"
Xuat_max: .asciiz   "Phan tu lon nhat = "
Xuat_min: .asciiz   "Phan tu nho nhat = "
#---------------------------------
# Code segment
     .text
     .globl    main
#---------------------------------
# Chuong trinh chinh
#---------------------------------
main:
# Nhap (syscall)
  # goi nhap_pt
     la   $a0,flo_arr
     lw   $a1,int_n
     jal  nhap_pt
# Xu ly
  # goi max_day
     la   $a0,flo_arr
     lw   $a1,int_n
     jal  max_day
  # luu max
     swc1 $f0,flo_max
  # goi min_day
     la   $a0,flo_arr
     lw   $a1,int_n
     jal  min_day
  # luu min
     swc1 $f0,flo_min
# Xuat ket qua (syscall)
  # goi xuat_pt
     la   $a0,flo_arr
     lw   $a1,int_n
     jal  xuat_pt
  # xuong dong
     addi $a0,$zero,'\n'
     addi $v0,$zero,11
     syscall
  # xuat max
     la   $a0,Xuat_max
```

```
      addi $v0,$zero,4
      syscall
      lwc1 $f12,flo_max
      addi $v0,$zero,2
      syscall
  # xuong dong
      addi $a0,$zero,'\n'
      addi $v0,$zero,11
      syscall
  # xuat min
      la   $a0,Xuat_min
      addi $v0,$zero,4
      syscall
      lwc1 $f12,flo_min
      addi $v0,$zero,2
      syscall
# Ket thuc chuong trinh (syscall)
Kthuc:    addiu     $v0,$zero,10
      syscall
#-----------------------------------
# ham nhap_pt: nhap phan tu day
# In: a0=addr(a[]), a1=spt n
# Out: none
# Reserved: none
#-----------------------------------
nhap_pt:
  # a2=addr(a[])
      add  $a2,$a0,$zero
  # xuat cau nhac
      la   $a0,Nhap_day
      addi $v0,$zero,4
      syscall
  # s0=i(=0)
  # for1-init
      addi $s0,$zero,0          #i=0
  # cond1
cond1:    beq  $s0,$a1,endfor1      #kiem tra (i==n)
  # bodyf1
    # xuat cau nhac va nhap
      la   $a0,Nhap_pt1
      addi $v0,$zero,4
      syscall
      add  $a0,$s0,$zero        # chi so i
      addi $v0,$zero,1
      syscall
      la   $a0,Nhap_pt2
      addi $v0,$zero,4
      syscall
      addi $v0,$zero,6          # nhap so thuc
      syscall
      swc1 $f0,0($a2)           # luu a[i]
  # loop1
      addi $s0,$s0,1
      addi $a2,$a2,4
```

```
     j    cond1
  # endfor1
endfor1:
     jr   $ra
#---------------------------------
# ham xuat_pt: xuat phan tu day
# In: a0=addr(a[]), a1=spt n
# Out: none
# Reserved: none
#---------------------------------
xuat_pt:
  # a2=addr(a[])
     add  $a2,$a0,$zero
  # xuat cau nhac
     la   $a0,Xuat_day
     addi $v0,$zero,4
     syscall
  # s0=i(=0)
  # for2-init
     addi $s0,$zero,0
  # cond2
cond2:    beq  $s0,$a1,endfor2
  # bodyf2
     lwc1 $f12,0($a2)         # xuat a[i]
     addi $v0,$zero,2
     syscall
    # ky tu TAB
     addi $a0,$zero,'\t'
     addi $v0,$zero,11
     syscall
  # loop2
     addi $s0,$s0,1
     addi $a2,$a2,4
     j    cond2
  # endfor2
endfor2:
     jr   $ra
#---------------------------------
# ham max_day: tim phan tu lon nhat
# In: a0=addr(a[]), a1=spt n
# Out: f0=pt max
# Reserved: none
#---------------------------------
max_day:
  # a0=addr(a[]), f0=a[0]/max, f1=a[i], s0=i(=1)
     lwc1 $f0,0($a0)          # max=a[0]
  # for3-init
     addi $s0,$zero,1         #i=1
     addi $a0,$a0,4           #addr(a[1])
  # cond3
cond3:    beq  $s0,$a1,endfor3
  # bodyf3
     lwc1 $f1,0($a0)          #a[i]
    # if3 (a[i]>max)
```

```
     c.le.s    $f1,$f0        #kiem tra (a[i]<=max)
     bc1t endif3              #dung, khong cap nhat max
    # then3 max=a[i]
     mov.s     $f0,$f1
    # endif3
endif3:
  # loop3
     addi $s0,$s0,1
     addi $a0,$a0,4
     j    cond3
  # endfor3
endfor3:                      # tri tra ve trong f0
     jr   $ra
#---------------------------------
# ham min_day: tim phan tu nho nhat
# In: a0=addr(a[]), a1=spt n
# Out: f0=pt min
# Reserved: none
#---------------------------------
min_day:
  # a0=addr(a[]), f0=a[0]/min, f1=a[i], s0=i(=1)
     lwc1 $f0,0($a0)
  # for4-init
     addi $s0,$zero,1
     addi $a0,$a0,4
  # cond4
cond4:    beq  $s0,$a1,endfor4
  # bodyf4
     lwc1 $f1,0($a0)
    # if4 (a[i]<min)
     c.lt.s    $f1,$f0        #kiem tra (a[i]<min)
     bc1f endif4              #sai, khong cap nhat
    # then4 min=a[i]
     mov.s     $f0,$f1
    # endif4
endif4:
  # loop4
     addi $s0,$s0,1
     addi $a0,$a0,4
     j    cond4
  # endfor4
endfor4:                      # tri tra ve trong f0
     jr   $ra
#---------------------------------
```