```
# Chuong trinh: C31B2 chu vi/dien tich hinh tron
# cv=2*PI*r, dt=PI*r*r
#-----------------------------------
# Data segment
     .data
# Cac dinh nghia bien
flo_r:    .float    1.3
flo_cv:   .float    2.7
flo_dt:   .float    3.5
flo_PI:   .float    3.141592
# Cac cau nhac nhap du lieu
Nhap_r:   .asciiz   "Nhap ban kinh r: "
Xuat_cv:  .asciiz   "Chu vi =      "
Xuat_dt:  .asciiz   "Dien tich =   "
Xuat_bka: .asciiz   "Ban kinh am hoac bang khong!"
#-----------------------------------
# Code segment
     .text
     .globl    main
#-----------------------------------
# Chuong trinh chinh
#-----------------------------------
main:
# Nhap (syscall)
  # Nhap ban kinh r
     la   $a0,Nhap_r
     addi $v0,$zero,4
     syscall
     addi $v0,$zero,6
     syscall
     swc1 $f0,flo_r
  # if (r>0)
     mtc1 $zero,$f1
     c.le.s    $f0,$f1   # kiem tra (r<=0)
     bc1t In_bka    # re nhanh neu dung
  # then : tinh chu vi/dien tich
# Xu ly
  # f0=cv/dt, f1=r, f2=PI, f3=2.0
     lwc1 $f1,flo_r
     lwc1 $f2,flo_PI
    # f3=2.0 cach 2 (doi ra dang luu tru IEEE 754)
     lui  $t0,0x4000
     ori  $t0,$t0,0x0000
     mtc1 $t0,$f3        #f3=2.0
    # f3=2.0 cach 3 (dung cho so nguyen)
     addi $t0,$zero,2    # nap so nguyen vao rd
     mtc1 $t0,$f3        # chuyen sang th/gh $f
     cvt.s.w   $f3,$f3   # doi so nguyen ra thuc trong $f
  # cv=2*PI -> cv=cv*r
     mul.s     $f0,$f3,$f2
     mul.s     $f0,$f0,$f1
     swc1 $f0,flo_cv
  # dt=PI*r -> dt=dt*r
     mul.s     $f0,$f2,$f1
```

```
     mul.s     $f0,$f0,$f1
     swc1 $f0,flo_dt
# Xuat ket qua (syscall)
  # Xuat chu vi
     la   $a0,Xuat_cv
     addi $v0,$zero,4
     syscall
     lwc1 $f12,flo_cv
     addi $v0,$zero,2
     syscall
  # xuong dong
     addi $a0,$zero,'\n'
     addi $v0,$zero,11
     syscall
  # Xuat dien tich
     la   $a0,Xuat_dt
     addi $v0,$zero,4
     syscall
     lwc1 $f12,flo_dt
     addi $v0,$zero,2
     syscall
     j    Kthuc
   # else
In_bka:   la   $a0,Xuat_bka
     addi $v0,$zero,4
     syscall
# Ket thuc chuong trinh (syscall)
Kthuc:    addiu     $v0,$zero,10
     syscall
#-----------------------------------
```